
	Reason for change:
	Current TTCN implementation will result in "basicServiceGroupList" field to be initialized with an incomplete value "{ omit }" in case p_BasicServiceCode is not specified or "omit" indicating the absence of an optional field is used during template reference.
Such value of cs_SS_Data() cannot be used for sending and will result in a runtime error.

  template (value) SS_Info cs_SS_Data(SS_Code p_SS_Code,
                                      template (omit) BasicServiceCode p_BasicServiceCode := omit,
                                      template (omit) SS_Status p_SS_Status := omit) :=
...
      basicServiceGroupList :=  { p_BasicServiceCode },
...

As according to specs:

Omitted optional record field value will result in runtime error as per "ETSI ES 201 873-1" clause "15.1 Declaring message templates"
At the time of a send operation, the used template shall be completely initialized and all fields shall resolve to
actual values or to omit and no other matching mechanisms shall be used in the template fields, neither directly
nor indirectly.

in combination with clause "15.7.2 Special symbols that can be used instead of values"
omit can be assigned to templates of any type as a whole or to optional fields of record and set types.
omit can only be used for matching optional fields.

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc64052562]Corrections required
0. [bookmark: _Toc64052563]Correction to template cs_SS_Data
	Template name
	cs_SS_Data

	Reason for change
	Current TTCN implementation will result in "basicServiceGroupList" field to be initialized with an incomplete value "{ omit }" in case p_BasicServiceCode is not specified or "omit" indicating the absence of an optional field is used during template reference.
Such value of cs_SS_Data() cannot be used for sending and will result in a runtime error.

	Summary of change
	Modifyed cs_SS_Data input parameter type as BasicServiceGroupList instead type BasicServiceCode, one element of the list, this way the list of elements can be assigned to omit as a whole.

	TTCN module
	Common/SS_CommonFunctions.ttcn

	MCC160 Comment
	




Before Change:
	<<SKIPPED CODE>>
  template (value) SS_Info cs_SS_Data(SS_Code p_SS_Code,
                                      template (omit) BasicServiceCode p_BasicServiceCode := omit,
                                      template (omit) SS_Status p_SS_Status := omit) :=
  { /* @status    APPROVED (SSNITZ) */
    // @sic R5s160389 sic@
    ss_Data := {
      ss_Code := p_SS_Code,
      ss_Status := p_SS_Status,
      ss_SubscriptionOption := omit,
      basicServiceGroupList :=  { p_BasicServiceCode },
      defaultPriority       := omit,
      nbrUser               := omit
    }
   };
<<SKIPPED CODE>>

	




After Change:
	<<SKIPPED CODE>>
  template (value) SS_Info cs_SS_Data(SS_Code p_SS_Code,
                                      template (omit) BasicServiceGroupList p_BasicServiceGroupList := omit,
                                      template (omit) SS_Status p_SS_Status := omit) :=
  { /* @status    APPROVED (SSNITZ) */
    // @sic R5s160389 sic@
    ss_Data := {
      ss_Code := p_SS_Code,
      ss_Status := p_SS_Status,
      ss_SubscriptionOption := omit,
      basicServiceGroupList := p_BasicServiceGroupList,
      defaultPriority       := omit,
      nbrUser               := omit
    }
   };
<<SKIPPED CODE>>





0. [bookmark: _Toc64052564]Correction to function f_TC_15_8_6_UTRAN
	Function name
	f_TC_15_8_6_UTRAN

	Reason for change
	Current TTCN implementation will result in "basicServiceGroupList" field to be initialized with an incomplete value "{ omit }" in case p_BasicServiceCode is not specified or "omit" indicating the absence of an optional field is used during template reference.
Such value of cs_SS_Data() cannot be used for sending and will result in a runtime error.

	Summary of change
	Adapted test case to use input parameter for cs_SS_Data as per Change 1.1.

	TTCN module
	Common/SS_CommonFunctions.ttcn

	MCC160 Comment
	




Before Change:
	<<SKIPPED CODE>>
  function f_TC_15_8_6_UTRAN() runs on UTRAN_PTC
  { /* Call restriction - Deactivation accepted */
    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;
    var Components v_FacilityInvokeBarring;
    var Components v_FacilityResult_Password;
    var octetstring v_FacilityReturn;
    var octetstring v_FacilityPassword;
    var InvokeIdType v_InvokeID_Password;
    var RRC_DATA_IND v_RRC_DataInd;
    var TI v_TI_R := f_NAS_TI_Init(cs_TI_MT);
    var TI v_TI_S;
    
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    
    // The UE is in registered idle state on cell 1
    f_UTRAN34_Preamble(utran34_Cell1);
    f_UTRAN_TestBody_Set(true);
    
    // @siclog "Step 1" siclog@
    // The UE is made to initiate the deactivation of all supplementary services restriction, for speech
    f_UT_SupplementaryService_CallBarring(UT,
                                          DEACTIVATE_MODE,
                                          SS_ALL_BARRING_AB,
                                          tsc_SS_Password,
                                          SS_SPEECH);
    // @siclog "Steps 2-4" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeDeactivate(?, cr_SS_ForBS_Code(tsc_SS_Code_AllBarringSS, (cr_Teleservice_AllSpeech, cr_Teleservice_TS11))), // @sic R5-145077 sic@
                                                    "Step 2",
                                                    "Step 4",
                                                    true);
    v_FacilityInvokeBarring := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
    
    // Set TI value for response to be sent from the SS
    v_TI_S := f_NAS_TI_InitAndSetFlag(v_TI_R, '1'B);
    
    // @siclog "Step 5" siclog@
    // FACILITY - Invoke = GetPassword (GuidanceInfo = enterPW).
    // The password for the relevant supplementary service (all supplementary services, or outgoing calls, respectively) is requested
    v_InvokeID_Password := v_FacilityInvokeBarring[0].invoke.invokeID + 1;
    
    v_FacilityPassword := f_SS_EncodeComponents(cs_Components(cs_InvokeComponent(cs_InvokeGetPassword(v_InvokeID_Password,
                                                                                                      v_FacilityInvokeBarring[0].invoke.invokeID,
                                                                                                      enterPW))));
    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_Facility(v_TI_S, cs_FacilityIE_lv(v_FacilityPassword))));
    
    // @siclog "Step 6" siclog@
    // FACILITY - Return result = GetPassword (<password>)
    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,
                                          tsc_RB3,
                                          cr_U_Facility(v_TI_R, cr_FacilityIE_lv))) -> value v_RRC_DataInd;
    v_FacilityResult_Password := f_SS_ComponentsDecode(v_RRC_DataInd.msg.facilityul.facility_Str.components); // @sic R5s140989 sic@
    if (not match(v_FacilityResult_Password, cr_Components(cr_ReturnResult(cr_ResultGetPassword(v_InvokeID_Password))))) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Facility message does not match to template");
    } else {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    }
    
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(cs_ReturnResultDeactivate(v_FacilityInvokeBarring[0].invoke.invokeID,
                                                                                                         cs_SS_Data(tsc_SS_Code_AllBarringSS, v_FacilityInvokeBarring[0].invoke.parameter.deactivateSS.basicService))))); // @sic R5-145077 sic@
    
    // @siclog "Step 7" siclog@
    // SS sends RELEASE COMPLETE with ActivateSS operation Return_result
    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_ReleaseCompleteFacility(v_TI_S, cs_FacilityIE_tlv(v_FacilityReturn))));
    
<<SKIPPED CODE>>

	




After Change:
	<<SKIPPED CODE>>
  function f_TC_15_8_6_UTRAN() runs on UTRAN_PTC
  { /* Call restriction - Deactivation accepted */
    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;
    var Components v_FacilityInvokeBarring;
    var Components v_FacilityResult_Password;
    var octetstring v_FacilityReturn;
    var octetstring v_FacilityPassword;
    var InvokeIdType v_InvokeID_Password;
    var RRC_DATA_IND v_RRC_DataInd;
    var TI v_TI_R := f_NAS_TI_Init(cs_TI_MT);
    var TI v_TI_S;
    var template (omit) BasicServiceGroupList v_BasicServiceGroupList := omit;
    
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    
    // The UE is in registered idle state on cell 1
    f_UTRAN34_Preamble(utran34_Cell1);
    f_UTRAN_TestBody_Set(true);
    
    // @siclog "Step 1" siclog@
    // The UE is made to initiate the deactivation of all supplementary services restriction, for speech
    f_UT_SupplementaryService_CallBarring(UT,
                                          DEACTIVATE_MODE,
                                          SS_ALL_BARRING_AB,
                                          tsc_SS_Password,
                                          SS_SPEECH);
    // @siclog "Steps 2-4" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeDeactivate(?, cr_SS_ForBS_Code(tsc_SS_Code_AllBarringSS, (cr_Teleservice_AllSpeech, cr_Teleservice_TS11))), // @sic R5-145077 sic@
                                                    "Step 2",
                                                    "Step 4",
                                                    true);
    v_FacilityInvokeBarring := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
    
    // Set TI value for response to be sent from the SS
    v_TI_S := f_NAS_TI_InitAndSetFlag(v_TI_R, '1'B);
    
    // @siclog "Step 5" siclog@
    // FACILITY - Invoke = GetPassword (GuidanceInfo = enterPW).
    // The password for the relevant supplementary service (all supplementary services, or outgoing calls, respectively) is requested
    v_InvokeID_Password := v_FacilityInvokeBarring[0].invoke.invokeID + 1;
    
    v_FacilityPassword := f_SS_EncodeComponents(cs_Components(cs_InvokeComponent(cs_InvokeGetPassword(v_InvokeID_Password,
                                                                                                      v_FacilityInvokeBarring[0].invoke.invokeID,
                                                                                                      enterPW))));
    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_Facility(v_TI_S, cs_FacilityIE_lv(v_FacilityPassword))));
    
    // @siclog "Step 6" siclog@
    // FACILITY - Return result = GetPassword (<password>)
    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,
                                          tsc_RB3,
                                          cr_U_Facility(v_TI_R, cr_FacilityIE_lv))) -> value v_RRC_DataInd;
    v_FacilityResult_Password := f_SS_ComponentsDecode(v_RRC_DataInd.msg.facilityul.facility_Str.components); // @sic R5s140989 sic@
    if (not match(v_FacilityResult_Password, cr_Components(cr_ReturnResult(cr_ResultGetPassword(v_InvokeID_Password))))) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Facility message does not match to template");
    } else {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    }
    
    if ( ispresent(v_FacilityInvokeBarring[0].invoke.parameter.deactivateSS.basicService))
    {
        v_BasicServiceGroupList := { v_FacilityInvokeBarring[0].invoke.parameter.deactivateSS.basicService};
    } 
      
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(cs_ReturnResultDeactivate(v_FacilityInvokeBarring[0].invoke.invokeID,
                                                                                                         cs_SS_Data(tsc_SS_Code_AllBarringSS, v_BasicServiceGroupList))))); // @sic R5-145077 sic@
    
    // @siclog "Step 7" siclog@
    // SS sends RELEASE COMPLETE with ActivateSS operation Return_result
    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_ReleaseCompleteFacility(v_TI_S, cs_FacilityIE_tlv(v_FacilityReturn))));
    
<<SKIPPED CODE>>


0. [bookmark: _Toc64052565]Correction to function f_TC_15_5_7_UTRAN
	Function name
	f_TC_15_5_7_UTRAN

	Reason for change
	Current TTCN implementation will result in "basicServiceGroupList" field to be initialized with an incomplete value "{ omit }" in case p_BasicServiceCode is not specified or "omit" indicating the absence of an optional field is used during template reference.
Such value of cs_SS_Data() cannot be used for sending and will result in a runtime error.

	Summary of change
	Adapted test case to use input parameter for cs_SS_Data as per Change 1.1

	TTCN module
	Common/SS_CommonFunctions.ttcn

	MCC160 Comment
	




Before Change:
	<<SKIPPED CODE>>
  function f_TC_15_5_7_UTRAN() runs on UTRAN_PTC
  { /* Call completion supplementary services, Activation */
    //@sic R5-169007 sic@
    var Components v_FacilityInvoke;
    var octetstring v_FacilityReturn;
    var TI v_TI_R := valueof(cs_TI_MT);
    var TI v_Ti_S;
    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;
    var boolean v_DoAuth := true;
    var template (value) ReturnResult v_ReturnResultActivate;
    
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    f_UTRAN34_Preamble(utran34_Cell1);
    
    f_UTRAN_TestBody_Set(true);
    
    // @siclog "Step 1" siclog@
    // The UE is made to initiate activation of call waiting for a basic service group speech.
    f_UT_SupplementaryService_CallWaiting(UT, ACTIVATE_MODE, SS_SPEECH);
    
    // @siclog "Steps 2-4" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeActivate(?, cr_SS_ForBS_Code(?,?)),  //@sic R5s150593 sic@
                                                    "-",
                                                    "-",
                                                    v_DoAuth);
    v_DoAuth := false;
    v_FacilityInvoke := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
    
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_ReturnResultActivate := cs_ReturnResultActivate(v_FacilityInvoke[0].invoke.invokeID,
                                                      cs_SS_Data(v_FacilityInvoke[0].invoke.parameter.activateSS.ss_Code, // @sic R5s160389 sic@
                                                                 v_FacilityInvoke[0].invoke.parameter.activateSS.basicService,
                                                                 omit));
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(v_ReturnResultActivate)));
    
    // Set TI value for response to be sent from the SS
    v_Ti_S := v_TI_R;
    v_Ti_S.tiFlag := '1'B;
      
    // @siclog "Step 5" siclog@
    // SS sends RELEASE COMPLETE with the ReturnResult
    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_ReleaseCompleteFacility(v_Ti_S, cs_FacilityIE_tlv(v_FacilityReturn))));

    // @siclog "Step 5A" siclog@
    // The SS releases the RRC connection
    f_UTRAN34_RRC_ConnRelStatus(utran34_Cell1);
    
    // @siclog "Step 6 " siclog@
    // Provide user indication
    if (f_UT_ChkSuppServiceUserIndication(UT, "", "Successful activation of call waiting service for a basic service group speech")) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
    }
    // @siclog "Step 7 - Void" siclog@

    // @siclog "Step 8" siclog@
    // The UE is made to initiate activation of call waiting for all basic service groups supported by the UE
    f_UT_SupplementaryService_CallWaiting(UT, ACTIVATE_MODE);
    
    // @siclog "Steps 9-11" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeActivate(?, cr_SS_ForBS_Code(?)),
                                                    "-",
                                                    "-",
                                                    v_DoAuth);
    v_FacilityInvoke := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
    
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_ReturnResultActivate := cs_ReturnResultActivate(v_FacilityInvoke[0].invoke.invokeID,
                                                      cs_SS_Data(v_FacilityInvoke[0].invoke.parameter.activateSS.ss_Code, // @sic R5s160389 sic@
                                                                 v_FacilityInvoke[0].invoke.parameter.activateSS.basicService,
                                                                 omit));
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(v_ReturnResultActivate)));
<<SKIPPED CODE>>

	




After Change:
	<<SKIPPED CODE>>
  function f_TC_15_5_7_UTRAN() runs on UTRAN_PTC
  { /* Call completion supplementary services, Activation */
    //@sic R5-169007 sic@
    var Components v_FacilityInvoke;
    var octetstring v_FacilityReturn;
    var TI v_TI_R := valueof(cs_TI_MT);
    var TI v_Ti_S;
    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;
    var boolean v_DoAuth := true;
    var template (value) ReturnResult v_ReturnResultActivate;
	var template (omit) BasicServiceGroupList v_BasicServiceGroupList; 
	   
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    f_UTRAN34_Preamble(utran34_Cell1);
    
    f_UTRAN_TestBody_Set(true);
    
    // @siclog "Step 1" siclog@
    // The UE is made to initiate activation of call waiting for a basic service group speech.
    f_UT_SupplementaryService_CallWaiting(UT, ACTIVATE_MODE, SS_SPEECH);
    
    // @siclog "Steps 2-4" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeActivate(?, cr_SS_ForBS_Code(?,?)),  //@sic R5s150593 sic@
                                                    "-",
                                                    "-",
                                                    v_DoAuth);
    v_DoAuth := false;
    v_FacilityInvoke := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
      
    if ( ispresent(v_FacilityInvoke[0].invoke.parameter.activateSS.basicService)  )
    {
        v_BasicServiceGroupList := { v_FacilityInvoke[0].invoke.parameter.activateSS.basicService };
    }else{
        v_BasicServiceGroupList := omit;
    }     
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_ReturnResultActivate := cs_ReturnResultActivate(v_FacilityInvoke[0].invoke.invokeID,
                                                      cs_SS_Data(v_FacilityInvoke[0].invoke.parameter.activateSS.ss_Code, // @sic R5s160389 sic@
                                                                 v_BasicServiceGroupList,
                                                                 omit));
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(v_ReturnResultActivate)));
    
    // Set TI value for response to be sent from the SS
    v_Ti_S := v_TI_R;
    v_Ti_S.tiFlag := '1'B;
      
    // @siclog "Step 5" siclog@
    // SS sends RELEASE COMPLETE with the ReturnResult
    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_ReleaseCompleteFacility(v_Ti_S, cs_FacilityIE_tlv(v_FacilityReturn))));

    // @siclog "Step 5A" siclog@
    // The SS releases the RRC connection
    f_UTRAN34_RRC_ConnRelStatus(utran34_Cell1);
    
    // @siclog "Step 6 " siclog@
    // Provide user indication
    if (f_UT_ChkSuppServiceUserIndication(UT, "", "Successful activation of call waiting service for a basic service group speech")) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
    }
    // @siclog "Step 7 - Void" siclog@

    // @siclog "Step 8" siclog@
    // The UE is made to initiate activation of call waiting for all basic service groups supported by the UE
    f_UT_SupplementaryService_CallWaiting(UT, ACTIVATE_MODE);
    
    // @siclog "Steps 9-11" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeActivate(?, cr_SS_ForBS_Code(?)),
                                                    "-",
                                                    "-",
                                                    v_DoAuth);
    v_FacilityInvoke := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
    
    
    if ( ispresent(v_FacilityInvoke[0].invoke.parameter.activateSS.basicService)  )
    {
        v_BasicServiceGroupList := { v_FacilityInvoke[0].invoke.parameter.activateSS.basicService };
    }else{
        v_BasicServiceGroupList := omit;
    }  
      
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_ReturnResultActivate := cs_ReturnResultActivate(v_FacilityInvoke[0].invoke.invokeID,
                                                      cs_SS_Data(v_FacilityInvoke[0].invoke.parameter.activateSS.ss_Code, // @sic R5s160389 sic@
                                                                 v_BasicServiceGroupList,
                                                                 omit));
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(v_ReturnResultActivate)));
    
<<SKIPPED CODE>>



0. [bookmark: _Toc64052566]Correction to function f_TC_15_5_8_UTRAN
	Function name
	f_TC_15_5_8_UTRAN

	Reason for change
	Current TTCN implementation will result in "basicServiceGroupList" field to be initialized with an incomplete value "{ omit }" in case p_BasicServiceCode is not specified or "omit" indicating the absence of an optional field is used during template reference.
Such value of cs_SS_Data() cannot be used for sending and will result in a runtime error.

	Summary of change
	Adapted test case to use input parameter for cs_SS_Data as per Change 1.1

	TTCN module
	Common/SS_CommonFunctions.ttcn

	MCC160 Comment
	




Before Change:
	<<SKIPPED CODE>>
  function f_TC_15_5_8_UTRAN() runs on UTRAN_PTC
  { /* Call completion supplementary services, Deactivation */
    //@sic R5-169007 sic@
    var Components v_FacilityInvoke;
    var octetstring v_FacilityReturn;
    var TI v_TI_R := valueof(cs_TI_MT);
    var TI v_Ti_S;
    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;
    var boolean v_DoAuth := true;
    var template (value) ReturnResult v_ReturnResultDeactivate;

    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    f_UTRAN34_Preamble(utran34_Cell1);
    
    f_UTRAN_TestBody_Set(true);
    
    // @siclog "Step 1" siclog@
    // The UE is made to initiate deactivation of call waiting for a basic service group speech.
    f_UT_SupplementaryService_CallWaiting(UT, DEACTIVATE_MODE, SS_SPEECH);
    
    // @siclog "Steps 2-4" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeDeactivate(?, cr_SS_ForBS_Code(?,?)), //@sic R5s150595 sic@
                                                    "-",
                                                    "-",
                                                    v_DoAuth);
    v_FacilityInvoke := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
    v_DoAuth := false;
    
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_ReturnResultDeactivate := cs_ReturnResultDeactivate(v_FacilityInvoke[0].invoke.invokeID,
                                                          cs_SS_Data(v_FacilityInvoke[0].invoke.parameter.deactivateSS.ss_Code,  // @sic R5s160389 sic@
                                                                     v_FacilityInvoke[0].invoke.parameter.deactivateSS.basicService,
                                                                     omit));
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(v_ReturnResultDeactivate)));
    
    // Set TI value for response to be sent from the SS
    v_Ti_S := v_TI_R;
    v_Ti_S.tiFlag := '1'B;
      
    // @siclog "Step 5" siclog@
    // SS sends RELEASE COMPLETE with DeactivateSS operation Return result
    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_ReleaseCompleteFacility(v_Ti_S, cs_FacilityIE_tlv(v_FacilityReturn))));

    // @siclog "Step 5A" siclog@
    // The SS releases the RRC connection
    f_UTRAN34_RRC_ConnRelStatus(utran34_Cell1);
    
    // @siclog "Step 6 " siclog@
    // Provide user indication
    if (f_UT_ChkSuppServiceUserIndication(UT, "", "Successful deactivation of call waiting service for a basic service group speech")) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
    }
    
    // @siclog "Step 7 - Void" siclog@
    
    // @siclog "Step 8" siclog@
    // The UE is made to initiate deactivation of call waiting for all basic service groups supported by the UE
    f_UT_SupplementaryService_CallWaiting(UT, DEACTIVATE_MODE);
    
    // @siclog "Steps 9-11" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeDeactivate(?, cr_SS_ForBS_Code(?)),
                                                    "-",
                                                    "-",
                                                    v_DoAuth);
    v_FacilityInvoke := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
    
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_ReturnResultDeactivate := cs_ReturnResultDeactivate(v_FacilityInvoke[0].invoke.invokeID,
                                                          cs_SS_Data(v_FacilityInvoke[0].invoke.parameter.deactivateSS.ss_Code,  // @sic R5s160389 sic@
                                                                     v_FacilityInvoke[0].invoke.parameter.deactivateSS.basicService,
                                                                     omit));
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(v_ReturnResultDeactivate)));
    
<<SKIPPED CODE>>

	




After Change:
	<<SKIPPED CODE>>
  function f_TC_15_5_8_UTRAN() runs on UTRAN_PTC
  { /* Call completion supplementary services, Deactivation */
    //@sic R5-169007 sic@
    var Components v_FacilityInvoke;
    var octetstring v_FacilityReturn;
    var TI v_TI_R := valueof(cs_TI_MT);
    var TI v_Ti_S;
    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;
    var boolean v_DoAuth := true;
    var template (value) ReturnResult v_ReturnResultDeactivate;
    var template (omit) BasicServiceGroupList v_BasicServiceGroupList;
    
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    f_UTRAN34_Preamble(utran34_Cell1);
    
    f_UTRAN_TestBody_Set(true);
    
    // @siclog "Step 1" siclog@
    // The UE is made to initiate deactivation of call waiting for a basic service group speech.
    f_UT_SupplementaryService_CallWaiting(UT, DEACTIVATE_MODE, SS_SPEECH);
    
    // @siclog "Steps 2-4" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeDeactivate(?, cr_SS_ForBS_Code(?,?)), //@sic R5s150595 sic@
                                                    "-",
                                                    "-",
                                                    v_DoAuth);
    v_FacilityInvoke := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
    v_DoAuth := false;
    
    if ( ispresent(v_FacilityInvoke[0].invoke.parameter.deactivateSS.basicService)  )
    {
        v_BasicServiceGroupList := { v_FacilityInvoke[0].invoke.parameter.deactivateSS.basicService };
    }else{
        v_BasicServiceGroupList := omit;
    }
    
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_ReturnResultDeactivate := cs_ReturnResultDeactivate(v_FacilityInvoke[0].invoke.invokeID,
                                                          cs_SS_Data(v_FacilityInvoke[0].invoke.parameter.deactivateSS.ss_Code,  // @sic R5s160389 sic@
                                                                     v_BasicServiceGroupList,
                                                                     omit));
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(v_ReturnResultDeactivate)));
    
    // Set TI value for response to be sent from the SS
    v_Ti_S := v_TI_R;
    v_Ti_S.tiFlag := '1'B;
      
    // @siclog "Step 5" siclog@
    // SS sends RELEASE COMPLETE with DeactivateSS operation Return result
    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_ReleaseCompleteFacility(v_Ti_S, cs_FacilityIE_tlv(v_FacilityReturn))));

    // @siclog "Step 5A" siclog@
    // The SS releases the RRC connection
    f_UTRAN34_RRC_ConnRelStatus(utran34_Cell1);
    
    // @siclog "Step 6 " siclog@
    // Provide user indication
    if (f_UT_ChkSuppServiceUserIndication(UT, "", "Successful deactivation of call waiting service for a basic service group speech")) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
    }
    
    // @siclog "Step 7 - Void" siclog@
    
    // @siclog "Step 8" siclog@
    // The UE is made to initiate deactivation of call waiting for all basic service groups supported by the UE
    f_UT_SupplementaryService_CallWaiting(UT, DEACTIVATE_MODE);
    
    // @siclog "Steps 9-11" siclog@
    // CM SERVICE REQUEST - REGISTER
    v_ReceiveRegisterResult := f_SS_ReceiveRegister(utran34_Cell1,
                                                    cr_InvokeDeactivate(?, cr_SS_ForBS_Code(?)),
                                                    "-",
                                                    "-",
                                                    v_DoAuth);
    v_FacilityInvoke := v_ReceiveRegisterResult.components;
    v_TI_R := v_ReceiveRegisterResult.ti;
    
    if ( ispresent(v_FacilityInvoke[0].invoke.parameter.deactivateSS.basicService)  )
    {
        v_BasicServiceGroupList := { v_FacilityInvoke[0].invoke.parameter.deactivateSS.basicService };
    }else{
        v_BasicServiceGroupList := omit;
    }
    
    // Assign invokeID in the ReturnResult with the value received in the Invoke
    v_ReturnResultDeactivate := cs_ReturnResultDeactivate(v_FacilityInvoke[0].invoke.invokeID,
                                                          cs_SS_Data(v_FacilityInvoke[0].invoke.parameter.deactivateSS.ss_Code,  // @sic R5s160389 sic@
                                                                     v_BasicServiceGroupList,
                                                                     omit));
    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnResult(v_ReturnResultDeactivate)));
    
<<SKIPPED CODE>>



